
=LET( 

Получаем данные из таблицы по выбранному месяцу (ячейка B3) 

  data; FILTER( 

    'Данные из кабинета // По месяцам'!A2:M999; 

    'Данные из кабинета // По месяцам'!A2:A999 = B3 

  ); 

Убираем строки без названия кампании 

  cleanData; FILTER(data; INDEX(data;;2) <> ""); 

Заголовки итоговой таблицы 

  headers; { 

    "Кампания"\"ID кампании"\"Показы"\"CTR%"\"Клики"\"CPC ₽"\"Расходы ₽"\"%Отказов"\ 

    "CR% Кнопки"\"Кнопки"\"CPA Кнопки ₽"\"CR% Email"\"Email"\"CPA Email ₽"\ 

    "CR% Телефон"\"Телефон"\"CPA Телефон ₽"\"CR% Telegram"\"Telegram"\"CPA 
Telegram ₽"\ 

    "CR% WhatsApp"\"WhatsApp"\"CPA WhatsApp ₽"\"CR% Заявка"\"Заявка"\"CPA Заявка 
₽" 

  }; 

Преобразуем каждую строку cleanData 

  rows; BYROW(cleanData; LAMBDA(row; 

    LET( 

Извлекаем значения из строки 

      name; IFNA(INDEX(row; 2); 0); 

      id;   IFNA(INDEX(row; 3); 0); 

      imp;  IFNA(INDEX(row; 4); 0); 

      clk;  IFNA(INDEX(row; 5); 0); 

      cst;  IFNA(INDEX(row; 6); 0); 

      bnc;  IFNA(INDEX(row; 7); 0); 

      ld;   IFNA(INDEX(row; 8); 0); 

      wa;   IFNA(INDEX(row; 9); 0); 



      tg;   IFNA(INDEX(row; 10); 0); 

      eml;  IFNA(INDEX(row; 11); 0); 

      phn;  IFNA(INDEX(row; 12); 0); 

      btn;  IFNA(INDEX(row; 13); 0); 

Расчет метрик 

      ctr;       ROUND(IF(imp = 0; 0; clk / imp); 4); 

      cpc;       ROUND(IF(clk = 0; 0; cst / clk); 2); 

      br;        ROUND(IF(clk = 0; 0; bnc / clk); 4); 

      cr_btn;    ROUND(IF(clk = 0; 0; btn / clk); 4); 

      cpa_btn;   ROUND(IF(btn = 0; 0; cst / btn); 2); 

      cr_eml;    ROUND(IF(clk = 0; 0; eml / clk); 4); 

      cpa_eml;   ROUND(IF(eml = 0; 0; cst / eml); 2); 

      cr_phn;    ROUND(IF(clk = 0; 0; phn / clk); 4); 

      cpa_phn;   ROUND(IF(phn = 0; 0; cst / phn); 2); 

      cr_tg;     ROUND(IF(clk = 0; 0; tg / clk); 4); 

      cpa_tg;    ROUND(IF(tg = 0; 0; cst / tg); 2); 

      cr_wa;     ROUND(IF(clk = 0; 0; wa / clk); 4); 

      cpa_wa;    ROUND(IF(wa = 0; 0; cst / wa); 2); 

      cr_ld;     ROUND(IF(clk = 0; 0; ld / clk); 4); 

      cpa_ld;    ROUND(IF(ld = 0; 0; cst / ld); 2); 

Формируем строку 

      HSTACK( 

        name; id; imp; ctr; clk; cpc; cst; br; 

        cr_btn; btn; cpa_btn; 

        cr_eml; eml; cpa_eml; 

        cr_phn; phn; cpa_phn; 

        cr_tg; tg; cpa_tg; 

        cr_wa; wa; cpa_wa; 



        cr_ld; ld; cpa_ld 

      ) 

    ) 

  )); 

Извлекаем колонки для итогов 

  imp_all; INDEX(cleanData;;4); 

  clk_all; INDEX(cleanData;;5); 

  cst_all; INDEX(cleanData;;6); 

  bnc_all; INDEX(cleanData;;7); 

  ld_all;  INDEX(cleanData;;8); 

  wa_all;  INDEX(cleanData;;9); 

  tg_all;  INDEX(cleanData;;10); 

  eml_all; INDEX(cleanData;;11); 

  phn_all; INDEX(cleanData;;12); 

  btn_all; INDEX(cleanData;;13); 

Строка "ИТОГО" 

  total; { 

    "ИТОГО"\ 

    ""\ 

    SUM(imp_all)\ 

    ROUND( IF(SUM(imp_all)=0; 0; SUM(clk_all)/SUM(imp_all)); 4 )\ 

    SUM(clk_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/clk_all; clk_all<>0) ); 2 ); 0 )\ 

    ROUND(SUM(cst_all); 2 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(bnc_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(btn_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(btn_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/btn_all; btn_all<>0) ); 2 ); 0 )\ 



    IFERROR( ROUND( AVERAGE( FILTER(eml_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(eml_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/eml_all; eml_all<>0) ); 2 ); 0 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(phn_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(phn_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/phn_all; phn_all<>0) ); 2 ); 0 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(tg_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(tg_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/tg_all; tg_all<>0) ); 2 ); 0 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(wa_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(wa_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/wa_all; wa_all<>0) ); 2 ); 0 )\ 

    IFERROR( ROUND( AVERAGE( FILTER(ld_all/clk_all; clk_all<>0) ); 4 ); 0 )\ 

    SUM(ld_all)\ 

    IFERROR( ROUND( AVERAGE( FILTER(cst_all/ld_all; ld_all<>0) ); 2 ); 0 ) 

  }; 

Финальный результат: заголовки + строки + итого 

  VSTACK(headers; rows; total) 

) 


